HODGE DIAMONDS OF ADJOINT ORBITS

BRIAN CALLANDER AND ELIZABETH GASPARIM

ABSTRACT. We present a Macaulay2 package that produces compactifications of adjoint orbits and of the fibres of symplectic Lefschetz fibrations on them. We use Macaulay2 functions to calculate the corresponding Hodge diamonds, which then reveal topological information about the Lefschetz fibrations.

CONTENTS

1. HODGE DIAMONDS OF LEFSCHETZ FIBRATIONS VIA MACAULAY2

The package *ProjAdjoint* provides Macaulay2 algorithms for defining compactifications of adjoint orbits and of the fibres of symplectic Lefschetz fibrations (SLF) on them. The Macaulay2 function `hh` can then be used to calculate the corresponding Hodge diamonds. An SLF is a fibration $f: X \to \mathbb{C}$ that has only Morse type singularities such that the fibres of f are symplectic submanifolds of X outside the critical set, see [Se]. We need the topological information provided by Hodge diamonds to study categories of Lagrangian vanishing cycles on symplectic Lefschetz fibrations. These play an essential role in the Homological Mirror Symmetry conjecture [Ko], where such a category appears as the Fukaya category of a Landau-Ginzburg (LG) model. A *Landau-Ginzburg model* is a Kähler manifold X equipped with a holomorphic function $f: X \to \mathbb{C}$ called the *superpotential*. SLFs are nice examples of LG models. In fact, a rigorous definition of the Fukaya category is known only for SLFs and not in any greater generality.

Date: December 17, 2013.
We are grateful to Koushik Ray of the Department of Theoretical Physics, Indian Association For The Cultivation of Science, Kolkata, for running our large memory (34 GB) computations on their servers. We would also like to thank Daniel Grayson for the time he has generously spent assisting us with technical issues.

A recent theorem of [GGS1] showed the existence of the structure of SLFs on adjoint orbits of semisimple Lie algebras. These adjoint orbits are non-compact spaces. In fact, they are isomorphic to cotangent bundles of flag varieties [GGS2]. Since Macaulay2 calculates Hodge diamonds for compact varieties, one needs to to compactify these orbits for the calculations. Expressing the adjoint orbit as an algebraic variety, we homogenise its ideal to obtain a projective variety which serves as our compactification. We then obtain topological data for the total space X as well as for the fibres of f.

Remark 1. Choosing a compactification is in general a delicate task: a different choice of generators for the defining ideal of the orbit can result in completely different Hodge diamonds of the corresponding compactification. This happens because the homogenisation of an ideal I can change drastically if we vary the choice of generators for I. Our package always chooses the ideal generated by the entries of the minimal polynomial defining the matrices of the orbit.

2. Lefschetz fibrations on adjoint orbits

Let H_0 be an element in the Cartan subalgebra of a semisimple Lie algebra $\mathfrak{g}$, and let $\mathcal{O}(H_0)$ denote its adjoint orbit. It is proved in [GGS1] that for each regular element $H \in \mathfrak{g}$, the function $f_H : \mathcal{O}(H_0) \to \mathbb{C}$ given by $f_H(x) = \langle H, x\rangle$ gives the orbit the structure of a symplectic Lefschetz fibration.

We compactify the orbit by projectivisation, that is, we homogenise the polynomials by adding an extra variable t to obtain a projective variety.

3. Example: the case of $H_0 = \mathrm{Diag}(2,-1,-1) \in \mathfrak{sl}(3,\mathbb{C})$

In $\mathfrak{sl}(3,\mathbb{C})$, consider the orbit $\mathcal{O}(H_0)$ of

$$H_0 = \begin{pmatrix} 2 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & -1 \end{pmatrix}$$

under the adjoint action. We use Macaulay2 to calculate the Hodge diamonds of a compactification of the adjoint orbit $\mathcal{O}(H_0)$. We fix the element

$$H = \begin{pmatrix} 1 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

to define the potential f_H. The explicit formula for f_H is given below.

The function *compactOrbit* of *ProjAdjoint* takes one input: a list of numbers corresponding to the diagonal entries of a matrix $H_0 \in \mathfrak{sl}(n,\mathbb{C})$. It then outputs a projective variety which is a compactification of the adjoint orbit of H_0 in $\mathfrak{sl}(n,\mathbb{C})$.

The function *compactFibre* takes three inputs: two lists of numbers corresponding to the diagonal entries of matrices H and H_0 in $\mathfrak{sl}(n,\mathbb{C})$, and one complex number λ. The matrix H should be regular. The output is a projective variety which is a compactification of the fibre over λ.

3.1. **Orbit.** We must first define our base field $\Bbbk$ and ring of polynomials R. Note that the prime number 32749 is the largest prime that Macaulay2 can work with ([EGSS]).

```
i1 : k = ZZ/32749;

i2 : R = k[x_1, x_2, y_1..y_3, z_1..z_3, t];
```

The variable t will be used for projectivisation. A general element $A \in \mathfrak{sl}(3,\mathbb{C})$ has the form

```
i3 : A = matrix{
            {x_1, y_1, y_2},
            {z_1, x_2, y_3},
            {z_2, z_3, -x_1 - x_2}
     };

                3       3
o3 : Matrix R  <--- R
```

and we will also make use of the identity matrix

```
i4 : Id = id_(k^3);

                3       3
o4 : Matrix k  <--- k
```

In our example, the adjoint orbit $\mathcal{O}(H_0)$ consists of all the matrices with the minimal polynomial $(A+\mathrm{id})(A-2\,\mathrm{id})$, so we are interested in the variety cut out by the equation `minPoly`:

```
i6 : minPoly = (A + Id)*(A - 2*Id);

                3       3
o6 : Matrix R  <--- R

i7 : I = ideal minPoly;

o7 : Ideal of R
```

To obtain a projectivisation of X, we first homogenise its ideal I, then take the corresponding projective variety. Saturating the ideal does not change the projective variety it defines but can make computations faster.

```
i8 : Ihom = saturate homogenize(I,t);

o8 : Ideal of R

i9 : Xproj = Proj(R/Ihom);
```

One checks with the command `dim Xproj` that $\dim \overline{X} = 4$. To check that $\overline{X}$ is non-singular, we use:

```
i10 : codim singularLocus Ihom

o10 = 9
```

Since the codimension of the singularities is 9 but the dimension of the ambient projective space is 8, we deduce that our projective variety $\overline{X}$ must be non-singular.

Now we calculate the Hodge diamond of $\overline{X}$. The Hodge numbers $h^{i,j}$ for $i+j \le 4$, $i \ge j$, are computed with the command `hh^(i,j) Xproj`. Since $\overline{X}$ is non-singular, the other entries of the Hodge diamond are given by the classical symmetries, as shown below.

$$
\begin{array}{ccccccccc}
 & & & & 1 & & & & \\
 & & & 0 & & 0 & & & \\
 & & 0 & & 2 & & 0 & & \\
 & 0 & & 0 & & 0 & & 0 & \\
0 & & 0 & & 3 & & 0 & & 0 \\
 & 0 & & 0 & & 0 & & 0 & \\
 & & 0 & & 2 & & 0 & & \\
 & & & 0 & & 0 & & & \\
 & & & & 1 & & & &
\end{array}
$$

3.2. **Regular fibre.** To define the regular and critical fibres, we also need the potential, which in our case is given by:

```
i20 : potential = x_1 - x_2;
```

The critical values of this potential are ± 3 and 0. Since all regular fibres of an SLF are isomorphic, it suffices to chose the regular value 1. We then define the regular fibre X_1 as the variety in $\mathfrak{sl}(3,\mathbb{C}) \cong \mathbb{C}^8$ corresponding to the ideal J:

```
i21 : J = ideal(minPoly) + ideal(potential-1);

o21 : Ideal of R
```

We then homogenise J to obtain a projectivisation $\overline{X}_1$ of the regular fibre X_1:

```
i22 : Jhom = saturate homogenize(J,t);

o22 : Ideal of R

i23 : X1proj = Proj(R/Jhom);
```

We check with the command `dim X1proj` that $\dim \overline{X}_1 = 3$. We use the command

```
i24 : codim singularLocus Jhom

o24 = 9
```

to test for singularities. Since this codimension is 9, we see that $\overline{X}_1$ is indeed non-singular. Now we calculate $h^{i,j}$ for $i+j \le 3$ e $i \ge j$ with the command `hh^(i,j) X1proj`. Since $\overline{X}_1$ is non-singular, the other entries of the Hodge diamond are obtained via the classical symmetries.

$$\begin{array}{ccccccc} & & & 1 & & & \\ & & 0 & & 0 & & \\ & 0 & & 2 & & 0 & \\ 0 & & 0 & & 0 & & 0 \\ & 0 & & 2 & & 0 & \\ & & 0 & & 0 & & \\ & & & 1 & & & \end{array}$$

Remark 2. We used the same method to calculate the Hodge diamonds for the singular fibre over 0 and obtained the same Hodge diamond as for the regular fibres.

Remark 3. More details of this example in appear in [B].

4. GENERALISATION: $H_0 = \mathrm{Diag}(n-1,-1,\ldots,-1) \in \mathfrak{sl}(n,\mathbb{C})$

We generalise our example of $\mathfrak{sl}(3,\mathbb{C})$ to $\mathfrak{sl}(n+1,\mathbb{C})$. To obtain the minimal flag, we set $H_0 = \mathrm{Diag}(n,-1,\ldots,-1)$ and $H = \mathrm{Diag}(1,-1,0,\ldots,0)$. Then the diffeomorphism type of the adjoint orbit is given by $\mathcal{O}(H_0) \simeq T^*\mathbb{P}^n$ (see [GGS2]), and H gives the potential $x_1 - x_2$ as before. If we compactify this orbit to $\mathbb{P}^n \times (\mathbb{P}^n)^*$, then the Hodge classes of the compactification are given by $h^{p,p} = n+1-|n-i|$ and the remaining Hodge numbers are 0. An application of the Lefschetz hyperplane theorem determines all but the Hodge numbers of the middle row of the compactification of the regular fibre.

5. COMPUTATIONAL COROLLARIES AND PITFALLS

The following two corollaries follow immediately from observing the Hodge diamonds we obtained.

Corollary 1. *Let $H_0 = \mathrm{Diag}(n-1,-1,\ldots,-1) \in \mathfrak{sl}(n,\mathbb{C})$ and $H = \mathrm{Diag}(1,-1,0,\ldots,0)$. Then the orbit compactifies holomorphically and symplectically to a trivial product.*

Proof. For the examples we considered here, [GGS2] showed that $\mathcal{O}(H_0)$ can be embedded differentiably into $\mathbb{P}^n \times \mathbb{P}^{n*}$. As an outcome of our computations, we verify that the compactifications are holomorphically and symplectically isomorphic to $\mathbb{P}^n \times \mathbb{P}^{n*}$ as well. In fact, our package produces a compactification of the orbit embedded into $\mathbb{P}^{n^2-1}$ and the diamond shows that the compactified orbit has the topological type of a $\mathbb{P}^n$ bundle over $\mathbb{P}^n$, these combined imply the bundle is trivial. □

Corollary 2. *An extension of the potential f_H to the compactification $\mathbb{P}^n \times \mathbb{P}^{n*}$ cannot be of Morse type, that is, it must have degenerate singularities.*

Proof. Our potential has singularities at $wH_0, w \in \mathcal{W}$. Now observe that the Hodge diamond of our compactified regular fibres have all zeroes in the middle row, hence any extension of the fibration to the compactification will have no vanishing cycles. However, the existence of a Lefschetz fibration with singularities and without vanishing cycles is prohibited by the fundamental theorem of Picard-Lefschetz theory. □

Remark 4 (Computational pitfalls). Macaulay2 greatly facilitates calculations of Hodge numbers that are unfeasible by hand. However, the memory requirements rise steeply with the dimension of the variety – especially for the Hodge classes $h^{p,p}$.

References

[B] B. Callander, *Lefschetz Fibrations*, Master's Thesis, Universidade Estadual de Campinas (2013).

[EGSS] D. Eisenbud, D. Grayson, M. Stillman, B. Sturmfels, *Computations in Algebraic Geometry with Macaulay2*, Springer (2002).

[GGS1] E. Gasparim, L. Grama, L. A. B. San Martin, *Lefschetz fibrations on adjoint orbits*, arXiv:1309.4418.

[GGS2] E. Gasparim, L. Grama, L. A. B. San Martin, *Adjoint orbits of semisimple Lie groups and Lagrangian submanifolds*, preprint.

[M2] D. Grayson, M. Stillman, *Macaulay2*, a software system for research in algebraic geometry, Available at http://www.math.uiuc.edu/Macaulay2/.

[Ko] M. Konsevich, *Homological algebra of Mirror Symmetry*, Proc. International Congress of Mathematicians (Zurich, 1994) Birkhäuser, Basel (1995) 120–139.

[Se] P. Seidel, *Fukaya categories and Picard-Lefschetz theory*, Zurich Lectures in Advanced Mathematics, European Math. Soc., Zurich (2008).